

MICRO-1マイクロ命令アセンブリ言語仕様

<マイクロプログラム> ::= <タイトル> <本体>

<タイトル> ::= ".TITLE" <名前>

<本体> ::= <マイクロ命令> { <マイクロ命令> }
".END"

<マイクロ命令> ::= <マイクロ命令頭部> <マイクロ命令本体>

<マイクロ命令頭部> ::= "*" <名札> <アドレス>

<名札> ::= <名前> ":" | <空>

<アドレス> ::= <符号なし整数> | <空>

<マイクロ命令本体> ::=
 <テストと順序制御文> // TS, SQ, LTフィールド
 <メモリ文> // MMフィールド
 <演算文> // LB, RB, AL, SH, SBフィールド
 <EX文> // EXフィールド

```
=====
* L0:      .TITLE      EXAMPLE      // プログラムの開始(タイトル)
          100          // マイクロ命令開始(ラベル&アドレス付)
          R2:=R1+R3     // 演算文
          C-1          // EX文
*          // マイクロ命令開始
          IF CRY=1 THEN L0 // テストと順序制御文
          R3:=SLL R4     // 演算文
          .END          // プログラムの終了
=====
```

<テストと順序制御文> ::= <goto文> | <call文> |
 <return文> | <if文> |
 <iop文> | <ira文> |
 <iab文> | <nsq文> | <空>

<goto文> ::= "GOTO" <名前> | // SQ=B, LT=アドレス
 "GOTO FETCH" // SQ=EI, TS=NTS

<call文> ::= "CALL" <名前> // SQ=BP, LT=アドレス

<return文> ::= "RETURN" // SQ=RTN

<if文> ::= "IF" <フラグ> "=1 THEN" <名前> |
 // SQ=BT, TS=テスト条件, LT=アドレス
 "IF" <フラグ> "=0 THEN" <名前> |
 // SQ=BF, TS=テスト条件, LT=アドレス
 "IF" <フラグ> "=0 THEN" <名前> "ELSE FETCH"
 // SQ=EI, TS=テスト条件, LT=アドレス

<フラグ> ::= "ZER" | "NEG" | "CRY" | "OV" | "CZ" | "T"

<iop文> ::= "IOP" <名前> // SQ=IOP, LT=アドレス

<ira文> ::= "IRA" <名前> // SQ=IRA, LT=アドレス

<irb文> ::= "IRB" <名前> // SQ=IRB, LT=アドレス

<nsq文> ::= "NSQ" // SQ=NSQ

<演算文> ::= <alu文> | <set文> | <空>

<alu文> ::= <sbus> ":" <式>

```

<set文> ::= "SET BY" <式>

<sbus> ::= "R0"|"R1"|"R2"|"R3"|"R4"|"R5"|"R6"|"R7"|
           "RA"|"RAP"|"RB"|"RBP"|"PC"
           // SBフィールドのニモニックに対応

<式> ::= <lbus> <aluとシフト> <rbus> | <alu素通り式>

<lbus> ::= "R0"|"R1"|"R2"|"R3"|"R4"|"R5"|"R6"|"R7"|
           "RB"|"RBP"|"PC"|"IO"|"MM"|"IR"|"FSR"|"ZERO"
           // LBフィールドのニモニックに対応

<aluとシフト> ::= <alu> ":" <シフト> | <alu>

<alu> ::= "+"|"-"|"AND"|"OR"|"XOR"|"$"|"@"
           // ALフィールドのニモニックに対応

<シフト> ::= "SLL"|"SRL"|"SLA"|"SRA"|"SNX"|"SWP"|"NSB"
           // SHフィールドのニモニックに対応

<rbus> ::= "R0"|"R1"|"R2"|"R3"|"R4"|"R5"|"R6"|"R7"|
           "RA"|"RAP" | <リテラル>
           // RBフィールドのニモニックに対応

<リテラル> ::= <符号なし整数(0～65536)>

<符号なし整数(0～65536)> ::= <16進数> | <10進数> | <2進数>

<16進数> ::= <16進数字> { <16進数字> }

<16進数字> ::= <数字> | "A"|"B"|"C"|"D"|"E"|"F"

<10進数> ::= 'D' <数字> { <数字> }

<2進数> ::= 'B' <2進文字> { <2進文字> }

<2進文字> ::= "0"|"1"

<数字> ::= "0"|"1"|"2"|"3"|"4"|"5"|"6"|"7"|"8"|"9"

<alu素通り式> ::= <シフト> <lbusまたはrbus> | <lbusまたはrbus>

<lbusまたはrbus> ::= <lbus> | <rbus>

<メモリ文> ::= "READ" |           // MM=RM
                "WRITE" |          // MM=WM
                <空>                // MM=NMM

<EX文> ::= "C-1"|"FLAG SAVE"|"WITH CRY"|"WITH ONE"|"IR:=" <lbusソース> |
           "IO:=" <lbusソース> | "C:=" <rbusソース> |
           "EXECUTE IO"|"T:=1"|"T:=0" |
           "IRA+1"|"IRB+1"|"IRB-1"|"SET HLT"|"SET OV"|"NEX" | <空>
           // EXフィールドのニモニックに対応

<lbusソース> ::= "LBUS" | <lbus>

<rbusソース> ::= "RBUS" | <rbus>

```

※ {}は0回以上の繰り返しを示す。

[eof]